

DAWSON COLLEGE  
COMPUTER SCIENCE DEPARTMENT

**LABORATORY PRESENTATION MANUAL**

## TABLE OF CONTENTS

		<b>Page</b>
I	Lab Report Contents .....	1
II	Lab Report Requirements .....	2
III	Sample Lab Report .....	3
IV	Report Program Guidelines .....	27

# I

## LAB REPORT CONTENTS

Your laboratory presentation should include the following detail specifications.

- 1 - Title page
- 2 - Table of contents
- 3 - Program narrative
- \* 4 - Program constraints (if any)
- 5 - System flowchart
- 6 - File description form(s)
- 7 - Printer Spacing Chart (or VDU/CRT form)
- 8 - Structure chart (Hierarchy chart)
- 9 - Structured flowchart or Pseudocode
- 10 - The test data file listing
- 11 - Validation of test results
- 12 - The Deskchecking Procedure
- 13 - The compiled listing of the source code
- 14 - The test data run
- 15 - The production run

### \*PROGRAM CONSTRAINTS:

In this section you state the limitations imposed on your program. For example, if you left a two digit field for hours worked, then the program can only handle hours worked up to 99 hours per pay period. Other constraints are such things as the maximum pay rate, gross pay, net pay, and sizes allocated to internal tables, if any. Impossible values are **not** listed as constraints. For example, if you leave a 3 digit field for the number of years that the employee has been with the company, it is physically impossible that this be considered a constraint. The reason being that you will never encounter an employee who has been with a company for over 999 years.

## II

### LAB REPORT REQUIREMENTS

1. All lab reports must be submitted in a **duo-tang** folder, with a label on the outside containing the following information:

Your name  
Title of the lab  
Lab Assignment #

2. All lab documentation must be produced using a word processor and checked for correct spelling, except where diagrams or hand calculations are involved.
3. Lab reports must be submitted on the date due, at the time requested. If lab reports are due at the start of class, those reports submitted during or after class will be considered late.

COBOL I  
420-201-90  
Section 02

LAB TITLE: Purchase Report

(Lab #1)

Prepared by: C.Legendre  
Student id: 9999999  
Date Due: 01/08/93

Prepared for: Teacher's name  
Date submitted:  
Grade Awarded:

TABLE OF CONTENTS

	<u>PAGE</u>
Program Narrative .....	5
System Flowchart .....	6
File Description Form .....	7
Printer Spacing Chart .....	8
Structure Chart .....	9
Structured Flowchart .....	10-15
Listing of Test Data File.....	16
Statement of Verification Technique.....	17
Deskchecking Routine .....	18-20
Compiled Listing .....	21-25
Test Run .....	25
Production Run .....	26

PROGRAM NARRATIVE

Program name: PURCHRPT

Purpose: The program prepares a detail report containing all customer purchases.

Description: Each record of the input file contains the following fields: a customer number, a customer name and a purchase amount.

For each customer record read an output line is printed containing the customer number, the customer name and the amount purchased by the customer. The data is printed in edited form. The program accumulates the purchase amounts so that the total amount purchased by the customers can be printed at the end of the report.

The output is a multi-page detail report with appropriate report and column headings. There is a maximum of 40 detail lines which can be printed per page.

Input file(s): A purchase transaction file. See attached File Description Form (page 7) and System Flow Chart (page 6).


Output: A purchase report file. See attached Printer  
File(s) Spacing Chart (page 8).

Program  
Constraints: Purchase amount cannot exceed \$99,999.99

Assumes a maximum of 100 customers in order to determine the size of the sum of the purchases.  
Maximum sum =  $99,999.99 \times 100 = \$9,999,999.99$

There is a maximum of 40 detail lines per page.

# SYSTEM FLOWCHART





# FILE DESCRIPTION FORM

**FILE DEFINITION:**

FILE NAME:  
 INTERNAL PURCHASE-TRANS-FILE  
 EXTERNAL PURCHASE.DAT (DOS FILE NAME/DDNAME)  
 DSNAME \_\_\_\_\_ (MAINFRAME ONLY)

MEDIUM: MICRO DISK X INPUT STREAM \_\_\_\_\_ MAINFRAME DISK \_\_\_\_\_

FILE TYPE:  
 MICRO: TEXT X MAINFRAME: PHYSICAL SEQUENTIAL \_\_\_\_\_  
 BINARY \_\_\_\_\_ VSAM-ESDS \_\_\_\_\_  
 VSAM-KSDS \_\_\_\_\_  
 VSAM-RRDS \_\_\_\_\_  
 PARTITIONED \_\_\_\_\_

**RECORD DEFINITION:**

RECORD NAME: PURCHASE-TRANS-REC  
 RECORD LENGTH: 32 BLOCKSIZE: NA  
 RECORD FORMAT: FIXED X  
 VARIABLE \_\_\_\_\_

PRIMARY KEY: NA KEY LENGTH: \_\_\_\_\_ KEY POSITION: \_\_\_\_\_  
 SECONDARY KEY: \_\_\_\_\_ KEY LENGTH: \_\_\_\_\_ KEY POSITION: \_\_\_\_\_

**FIELD DEFINITION(S):**


<u>COLUMNS</u>	<u>FIELD NAME</u>	<u>TYPE/PICTURE</u>	<u>SIZE/BYTES</u>
<u>01-05</u>	<u>CUST-NUM-IN</u>	<u>X(05)</u>	<u>5</u>
<u>06-25</u>	<u>CUST-NAME-IN</u>	<u>X(20)</u>	<u>20</u>
<u>26-32</u>	<u>PURCH-AMT-IN</u>	<u>9(05)V99</u>	<u>7</u>
_____	_____	_____	_____
_____	_____	_____	_____
_____	_____	_____	_____
_____	_____	_____	_____
_____	_____	_____	_____
_____	_____	_____	_____
_____	_____	_____	_____

FILE NAME: PURCHASE-RPT-FILE  
 RECORD LENGTH: 80 CHARACTERS


# PRINTER SPACING CHART


CARRIAGE CONTROL	1	2	3	4	5	6	7	8	9	0	1	2	3	4	5	6	7	8	9	0	1	2	3	4	5	6	7	8	9	0	1	2	3	4	5	6	7	8	9	0	1	2	3	4	5	6	7	8	9	0	1	2	3	4	5	6	7	8	9	0
H	1	MM/DD/YY																	PURCHASE REPORT																	PAGE 29																								
H	4	CUSTOMER NO																	CUSTOMER NAME																	AMOUNT OF PURCHASE																								
D	6	XXXXX																	XXXXXXXXXXXXXXXXXXXX																	ZZ,ZZ9.99																								
D	7	XXXXX																	XXXXXXXXXXXXXXXXXXXX																	ZZ,ZZ9.99																								
D	16	XXXXX																	XXXXXXXXXXXXXXXXXXXX																	ZZ,ZZ9.99																								
T	18																		TOTAL:																	\$8,899.99																								


STRUCTURE CHART





# STRUCTURED FLOWCHART
















TEST DATA LISTING

1234 Susan Smith  
21456 Tammy Cree  
34111 Peter Jones

0017523  
0021817  
0218319

TOTAL

0257659

## STATEMENT OF VERIFICATION TECHNIQUE

### Procedure:

1. Manually add the values in the field PURCH-AMT-IN in the test data file.
2. Deskcheck the flowchart. The total arrived at in step 1, will be compared with the total obtained in the deskchecking routine to ensure that the logic is correct.

3 ~~Total~~ totals will match Total on Test Run

# DESKCHECKING ROUTINE

CONSTANTS: MAX-LINES = 40  $\leftarrow$  (deskchecking routine assumes 2 lines per page)

CONTROL FLOW	EOF 'Y' END-OF-FILE	LINE-CTR	PG-CTR	PURCH-ACC	RUN-DATE	RT-DATE	CUST-NUM-IN	CUST-NAME-IN	PURCH-AMT-IN	RT-PG-CTR	CC
MM B1000	N	2	<del>X</del>	<del>X</del>	980123	01/23/92	12345	Susan Smith	0017523		
MM B2000											
C2000		3									
D2000		X	X							X	<del>X</del>
C2000											X
B2000				0017523			21456	Tammy Cree	0021817		
MM B2000											
C2000		<del>X</del>									X
B2000				0039340			34111	PETER Jones	0218319		
MM B2000											
C2000		3									
D2000		1								2	<del>X</del>
C2000			2								1
B2000	Y			0257659							
MM B3000											
MM											

18

DETAIL-LINE

CONTROL FLOW	DL-CUST- NUM	DL-CUST- NAME	DL-PURCH- AMT	TL-PURCH- ACC
C2000	<del>12345</del>	<del>Susan Smith</del>	<del>175.03</del>	
C2000	<del>21456</del>	<del>Tammy CREE</del>	218.17	
C2000	34111	PETER Jones	2183.19	
B3000				\$2,576.59

OUTPUT

01/23/92

Report title

PAGE 1

CUSTOMER NO

CUSTOMER NAME

AMOUNT OF PURCHASE

12345

Susan Smith

175.23

21456

Tammy Cree

218.17

01/23/92

REPORT TITLE

PAGE 2

CUSTOMER NO

CUSTOMER NAME

AMOUNT OF PURCHASE

34111

PETER JONES

2,183.19

TOTAL:

\$ 2,576.59

20

\* Options: LIST(PURCHRPT.LST) NOASMLIST

```
1 IDENTIFICATION DIVISION.  
2  
3 PROGRAM-ID. PURCHRPT.  
4 AUTHOR. C.LEGENDRE.  
5 DATE-WRITTEN. DECEMBER 14, 1993.  
6 DATE-COMPILED. 14-Dec-93 16:30.  
7  
8*****  
9* THIS PROGRAM PREPARES A PURCHASE DETAIL REPORT. *  
10* *  
11* EACH RECORD OF THE INPUT FILE CONTAINS THE FOLLOWING FIELDS: *  
12* A CUSTOMER NUMBER, A CUSTOMER NAME AND A PURCHASE AMOUNT. *  
13* *  
14* FOR EACH CUSTOMER RECORD READ AN OUTPUT LINE WILL BE PRINTED *  
15* CONTAINING THE CUSTOMER NUMBER, THE CUSTOMER NAME AND THE *  
16* AMOUNT PURCHASED BY THE CUSTOMER. THE DATA IS PRINTED IN *  
17* EDITED FORM. THE PROGRAM ACCUMULATES THE PURCHASE AMOUNTS *  
18* SO THAT THE TOTAL AMOUNT PURCHASED BY THE CUSTOMERS CAN BE *  
19* PRINTED AT THE END OF THE REPORT. *  
20* *  
21* THE OUTPUT IS A MULTI-PAGE DETAIL REPORT WITH APPROPRIATE *  
22* REPORT AND COLUMN HEADINGS. THERE IS A MAXIMUM OF 40 DETAIL *  
23* LINES PER PAGE. *  
24*****  
25  
26 ENVIRONMENT DIVISION.  
27  
28 INPUT-OUTPUT SECTION.  
29  
30 FILE-CONTROL.  
31 SELECT PURCHASE-TRANS-FILE ASSIGN TO "A:\PURCHASE.DAT"  
32 ORGANIZATION IS LINE SEQUENTIAL.  
33 SELECT PURCHASE-RPT-FILE ASSIGN TO "A:\PURCHRPT.PRN".  
34  
35 DATA DIVISION.  
36  
37 FILE SECTION.  
38  
39 FD PURCHASE-TRANS-FILE  
40 BLOCK CONTAINS 0 RECORDS  
41 RECORD CONTAINS 32 CHARACTERS.  
42  
43 01 PURCHASE-TRANS-REC.  
44  
45 05 CUST-NUM-IN PIC X(05).  
46 05 CUST-NAME-IN PIC X(20).  
47 05 PURCH-AMT-IN PIC 9(05)V99.  
48  
49 FD PURCHASE-RPT-FILE  
50 BLOCK CONTAINS 0 RECORDS  
51 RECORD CONTAINS 80 CHARACTERS.  
52  
53 01 OUTPUT-LINE PIC X(80).  
54  
55  
56 WORKING-STORAGE SECTION.  
57
```

```

58 01 REPORT-TITLE-LINE.
59
60     05 FILLER                PIC X(10) VALUE SPACES.
61     05 RT-DATE                PIC X(08).
62     05 FILLER                PIC X(13) VALUE SPACES.
63     05 FILLER                PIC X(30) VALUE 'PURCHASE REPORT'.
64     05 FILLER                PIC X(05) VALUE 'PAGE'.
65     05 RT-PG-CTR             PIC Z9.
66     05 FILLER                PIC X(12) VALUE SPACES.
67
68 01 COL-HDLINE.
69
70     05 FILLER                PIC X(10) VALUE SPACES.
71     05 FILLER                PIC X(14) VALUE 'CUSTOMER NO'.
72     05 FILLER                PIC X(26) VALUE 'CUSTOMER NAME'.
73     05 FILLER                PIC X(30) VALUE
74                               'AMOUNT OF PURCHASE'.
75
76 01 DETAIL-LINE.
77
78     05 FILLER                PIC X(13) VALUE SPACES.
79     05 DL-CUST-NUM           PIC X(05).
80     05 FILLER                PIC X(06) VALUE SPACES.
81     05 DL-CUST-NAME         PIC X(20).
82     05 FILLER                PIC X(15) VALUE SPACES.
83     05 DL-PURCH-AMT         PIC ZZ,ZZ9.99.
84     05 FILLER                PIC X(12) VALUE SPACES.
85
86 01 TOTAL-LINE.
87
88     05 FILLER                PIC X(47) VALUE SPACES.
89     05 FILLER                PIC X(08) VALUE 'TOTAL:'.
90     05 TL-PURCH-ACC         PIC $$$,$$$,$$9.99.
91     05 FILLER                PIC X(12) VALUE SPACES.
92
93
94 01 INDICATORS.
95
96     05 END-OF-FILE           PIC X(01) VALUE 'N'.
97     88 EOF                   VALUE 'Y'.
98     05 CC                    PIC 9(01).
99     05 MAX-LINES             PIC S9(02) VALUE +40 BINARY.
100
101 01 WORKFIELDS.
102
103     05 RUN-DATE.
104         10 RUN-YR            PIC 9(02).
105         10 RUN-MTH           PIC 9(02).
106         10 RUN-DAY           PIC 9(02).
107
108 01 COUNTERS.
109
110     05 LINE-CTR              PIC S9(02) VALUE +40 BINARY.
111     05 PG-CTR                PIC S9(02) VALUE +0 PACKED-DECIMAL.
112
113 01 ACCUMULATORS.
114
115     05 PURCH-ACC             PIC S9(07)V99 VALUE +0 PACKED-DECIMAL.
  
```



```
116
117 PROCEDURE DIVISION.
118
119 A0000-MAINLINE.
120
121     PERFORM B1000-INITIALIZE
122     PERFORM B2000-PROCESS
123         UNTIL EOF
124     PERFORM B3000-TERMINATE
125     STOP RUN.
126
127 B1000-INITIALIZE.
128
129*****
130* INVOKED BY THE MAINLINE RTN.  THIS MODULE OPENS THE FILES,      *
131* ACCEPTS THE RUN DATE AND DOES THE PRIME READ.                  *
132*****
133
134     OPEN INPUT PURCHASE-TRANS-FILE
135         OUTPUT PURCHASE-RPT-FILE.
136     ACCEPT RUN-DATE FROM DATE
137     STRING RUN-MTH '/' RUN-DAY '/' RUN-YR
138         DELIMITED BY SIZE INTO RT-DATE
139     READ PURCHASE-TRANS-FILE AT END
140     SET EOF TO TRUE
141     END-READ.
142
143 B2000-PROCESS.
144
145*****
146* INVOKED BY THE MAINLINE RTN.  THIS MODULE INVOKES THE DETAIL  *
147* LINE RTN IN ORDER TO SEND A LINE TO THE REPORT.  IT ACCUMULATES *
148* THE TOTAL PURCHASE AMOUNTS, THEN IT DOES THE SECONDARY READ.  *
149*****
150
151     PERFORM C2000-DETAIL-LINE
152     ADD PURCH-AMT-IN TO PURCH-ACC
153     READ PURCHASE-TRANS-FILE AT END
154     SET EOF TO TRUE
155     END-READ.
156
157 B3000-TERMINATE.
158
159*****
160* INVOKED BY THE MAINLINE RTN.  THIS MODULE PRINTS FINAL TOTALS *
161* AND CLOSES THE FILES.                                          *
162*****
163
164     MOVE PURCH-ACC TO TL-PURCH-ACC
165     WRITE OUTPUT-LINE FROM TOTAL-LINE
166         AFTER ADVANCING 2 LINES
167     CLOSE PURCHASE-TRANS-FILE
168         PURCHASE-RPT-FILE.
169
170 C2000-DETAIL-LINE.
171
172*****
173* INVOKED BY B2000-PROCESS.  THIS MODULE MOVES FIELDS TO THE  *

```

174\* DETAIL LINE, CHECKS FOR A PAGE OVERFLOW, AND PRINTS THE DETAIL \*  
175\* LINE. \*

176\*\*\*\*\*

177  
178 ADD 1 TO LINE-CTR  
179 IF LINE-CTR > MAX-LINES  
180 PERFORM D2000-HEADINGS  
181 END-IF  
182 MOVE CUST-NUM-IN TO DL-CUST-NUM  
183 MOVE CUST-NAME-IN TO DL-CUST-NAME  
184 MOVE PURCH-AMT-IN TO DL-PURCH-AMT  
185 WRITE OUTPUT-LINE FROM DETAIL-LINE  
186 AFTER ADVANCING CC LINES  
187 MOVE 1 TO CC.

188  
189 D2000-HEADINGS.

190  
191\*\*\*\*\*

192\* INVOKED BY C2000-DETAIL-LINE. THIS MODULE PRINTS THE REPORT \*  
193\* TITLES, RE-INITIALIZES THE LINE CTR AND SETS THE CARRIAGE CTRL.\*  
194\*\*\*\*\*

195

196 ADD 1 TO PG-CTR  
197 MOVE PG-CTR TO RT-PG-CTR  
198 WRITE OUTPUT-LINE FROM REPORT-TITLE-LINE  
199 AFTER ADVANCING PAGE  
200 WRITE OUTPUT-LINE FROM COL-HDLINE  
201 AFTER ADVANCING 3 LINES  
202 MOVE 1 TO LINE-CTR  
203 MOVE 2 TO CC.

CUSTOMER NO	CUSTOMER NAME	AMOUNT OF PURCHASE
12345	Susan Smith	175.23
21456	Tammy Cree	218.17
34111	Peter Jones	2,183.19
	TOTAL:	\$2,576.59

TEST RUN

CUSTOMER NO	CUSTOMER NAME	AMOUNT OF PURCHASE
12346	Susan Brown	900.00
13345	Thomas Armstrong	1,000.00
13555	James Atre	450.00
14567	Karen Bradley	100.00
14669	Linda Carlson	1,250.76
15678	Joanne Klimbie	837.50
16788	Robert Hardgrave	132.50
17890	Tammy Martin	210.40
23451	Claude Bouffard	901.23
23566	Douglas Clayton	1,245.60
23677	George Cliche	230.00
34566	Adolphus Lewis	457.75
45673	Caroline Reynolds	2,250.00
	TOTAL:	\$9,965.74

PRODUCTION RUN

COBOL I  
420-201-90  
Section 02

REPORT PROGRAM GUIDELINES

THE REPORT PROGRAM

Typical structural REPORT requirements

1. Headings - report  
- column
2. Detail lines
3. Summary lines
4. Pagination
5. Totals

1. HEADINGS

i) Report Headings

- appear at the top of each page of the report
- are usually centered on each page
- consist of a single line or two
- are simple and descriptive

Contents

- title of report  
declared as a literal in Working-Storage  
Section
- page number  
declared as a variable in Working-Storage  
Section
- date information presented  
taken from the system date  
(ACCEPT data-name FROM DATE)  
or user input

ii) Column Headings

- appear at the top of columns of data
- describe accurately the information below them
- well spaced for readability
- aligned with the data in the columns
  - a) centered for fixed length fields
  - b) left-justified for variable non-numeric fields
  - c) right-justified for numeric fields

2. Detail lines

- generally indicative or constant information is placed on the left and variable information on the right
- numeric data is right-justified
- non-numeric data is left justified
- detail information is edited before printing
- blank lines are inserted between report and column headings, and between column headings and summary lines

3. Summary lines

- totals of a group of records
- generally indicative or constant information is placed on the left and variable information on the right
- numeric data is right-justified
- non-numeric data is left-justified
- summary information is edited before printing
- blank lines are inserted between report and column headings, and between column headings and summary lines
- if the report contains both summary and detail lines then
  - . detail lines may be group indicated (key field(s) printed only when there is a change in group)
  - . blank lines are inserted between detail and summary lines

#### 4. Paqination

- titles are produced at start of report and whenever the maximum number of lines per page is reached
- line counter is required (declared as a variable in the Working-Storage Section)
- in Summary report, often a change in group will also force a new page

#### 5. Totals

- totals are the result of the summation of detail information in the columns
- totals are often printed within the same column as the data which is being summed; if so column width must be large enough to include the total field
- totals may be printed outside the column of detail information; however, an indicative literal should precede them
- totals are edited prior to printing

#### REPORT TYPES

1. Detail reports
2. Summary or Control break reports
3. Exception reports
4. Selection/Extraction reports
5. Classification reports

#### 1. Detail report

Each record in the input file produces one or more lines of output on the report.

Report recipient :

Lower management or clerical staff



Report contents:

Information reflecting the day-to-day operations - the operating status of the organization.

2. Summary or Control Break reports

The information is condensed. One or more lines of output is produced for each group of records in the input file. Accumulators (totals and subtotals) are printed.

Report recipient :

Lower, middle and higher management

Report contents :

Information that presents the overall picture of the organization, dept. etc.

3. Exception reports

One line or more of output is produced for each record that contains information that is "other than normal".

Report recipient :

Person requiring specific information to make decisions or to take specific actions

Report contents :

The required information at minimum cost and maximum usability.

EXAMPLES

A paycheck containing an excessive amount

An item in inventory whose quantity has fallen below its reorder point

4. Selection/Extraction reports

One or more lines of output are produced from each input record that satisfies predetermined criteria.

Report recipient :

Lower management or clerical staff

THE REPORT PROGRAM - PAGE 4

Report contents :

Similar to detail report, but list only those records which meet specified criteria - required information at minimum cost and maximum usability

5. Classification reports

Each record is classified according to specific criteria and then printed. One or more lines of output is produced for each input record read. The information must be sorted either on one or more groups prior to its being printed.

Report recipient :

Lower management or clerical staff

Report contents :

Similar to detail report, but breaks the report into two or more groups for processing.

EXAMPLES

A list by job title  
or years of experience

A list of grocery store sales by food items- perishable and  
non-perishable  
and non-food items





